Built-in Maple Routines for Finding Roots

solve is the general purpose equation solver in Maple. Let's apply it to

$$p(x) = x^5 + 11\,x^4 - 21\,x^3 - 10\,x^2 - 21\,x - 5\,.$$

We know there are real roots near -12, 0, and 2.

```
> restart;
> p:=x^5+11*x^4-21*x^3-10*x^2-21*x-5;
```

$$p := x^5 + 11\,x^4 - 21\,x^3 - 10\,x^2 - 21\,x - 5$$

```
> r:=solve(p=0,x);
```

$$r := \mathit{RootOf}\left(\_Z^5 + 11\,\_Z^4 - 21\,\_Z^3 - 10\,\_Z^2 - 21\,\_Z - 5, \mathit{index} = 1\right), \mathit{RootOf}\left(\_Z^5 + 11\,\_Z^4 - 21\,\_Z^3 - 10\,\_Z^2 - 21\,\_Z - 5, \mathit{index} = 2\right), \mathit{RootOf}\left(\_Z^5 + 11\,\_Z^4 - 21\,\_Z^3 - 10\,\_Z^2 - 21\,\_Z - 5, \mathit{index} = 3\right), \mathit{RootOf}\left(\_Z^5 + 11\,\_Z^4 - 21\,\_Z^3 - 10\,\_Z^2 - 21\,\_Z - 5, \mathit{index} = 4\right), \mathit{RootOf}\left(\_Z^5 + 11\,\_Z^4 - 21\,\_Z^3 - 10\,\_Z^2 - 21\,\_Z - 5, \mathit{index} = 5\right)$$

RootOf is a place holder for representing all the roots of an equation in one variable. It means the roots cannot be expressed exactly in terms of radicals. We try the following adjustment.

```
> evalf(r);
```

$$2.26008552806563, -0.198709531404488 + 0.813312546805162\,I, -0.250236940325127, -12.6124295249315, -0.198709531404488 - 0.813312546805162\,I$$

Let's look at a second example where there are exact roots.

```
> q:=4*x^3-3*x^2+2*x-12;
```

$$q := 4\,x^3 - 3\,x^2 + 2\,x - 12$$

```
> solve(q=0,x);
```

$$\frac{1}{12}\left(2511 + 12\sqrt{43809}\right)^{1/3} - \frac{5}{4\left(2511 + 12\sqrt{43809}\right)^{1/3}} + \frac{1}{4}, -\frac{1}{24}\left(2511 + 12\sqrt{43809}\right)^{1/3} + \frac{5}{8\left(2511 + 12\sqrt{43809}\right)^{1/3}} + \frac{1}{4} + \frac{1}{2}\,I\sqrt{3}\left(\frac{1}{12}\left(2511 + 12\sqrt{43809}\right)^{1/3} + \frac{5}{4\left(2511 + 12\sqrt{43809}\right)^{1/3}}\right), -\frac{1}{24}\left(2511 + 12\sqrt{43809}\right)^{1/3} + \frac{5}{8\left(2511 + 12\sqrt{43809}\right)^{1/3}} + \frac{1}{4} - \frac{1}{2}\,I\sqrt{3}\left(\frac{1}{12}\left(2511 + 12\sqrt{43809}\right)^{1/3} + \frac{5}{4\left(2511 + 12\sqrt{43809}\right)^{1/3}}\right)$$

These are three exact solutions. To get decimal approximations, use evalf.

```
> evalf(%);
```

$$1.604140225, -0.4270701129 + 1.299142722\,I, -0.4270701129 - 1.299142722\,I$$

fsolve uses numerical techniques to approximate roots. For a general equation, fsolve attempts to compute a single real root. However, for polynomials it will compute all real (non-complex) roots, although exceptionally ill-conditioned polynomials may cause fsolve to miss some roots. To compute all roots of a polynomial over the field of complex numbers, use the complex option. So it is important to look at the graph to get an idea of where the roots are. Let's go back to p.

```
> fsolve(p=0,x);
```

$$-12.61242952, -0.2502369403, 2.260085528$$

This just found the real roots.

```
> fsolve(p=0,x,complex);
```

$$-12.6124295249315, -0.250236940325127, -0.198709531404488 - 0.813312546805162 I, -0.198709531404488 + 0.813312546805162 I, 2.26008552806563$$

Suppose we just want to find the root near 2.2.

```
> fsolve(p=0,x=2..3);
```

$$2.260085528$$

We can also tell fsolve where to start looking for a root. This works like a first approximation in Newton's method.

```
> fsolve(sin(x)=0,x=3.1);
```

$$3.141592654$$